DETERMINATION OF OPTIMUM MAINTENANCE INTERVAL IN TERMS OF COST

CH. CHOWDHURY

Department of Precision Mechanics and Applied Optics,
Technical University, H-1521 Budapest

Received October 20, 1986
Presented by Prof. Dr. O. Petrik

Abstract

If a system can be characterized by an increasing failure rate, then it is advisable to replace it before it has aged too greatly. On the other hand frequent replacement means excessive cost. Therefore, the optimum replacement interval has to be determined which will minimize the maintenance cost. Here a simple method has been worked out to ascertain the optimum value of block replacement interval. An error committed in reference (1), while calculating the value of renewal density has been rectified. A necessary condition for calculating the optimum block interval is derived.

Introduction

In order to determine the preventive maintenance interval the expected cost for exchanging non-failed item (C_2) and the expected cost of failure (C_1) are required. In real life this value (the ratio of the cost of preventive maintenance and failure maintenance) may change in course of time. When this value changes, then the new preventive maintenance interval (T_0) is to be determined. In order to avoid repetition of the optimization and for the simplification of the consideration of the effect of the change of $\frac{C_2}{C_1}$, it will be useful if we can determine relationship between $\frac{C_2}{C_1}$ and T_0. In section 2 we develop a simple method to determine the block replacement interval and we also reveal a necessary conditions that is required for ascertaining the optimum block interval. In section 3 we develop and illustrate a diagram from which the value of T_0 can be readily seen if the value of $\frac{C_2}{C_1}$ is known.

Determination of block time

Under a block replacement, the unit is replaced at times $KT(K-1, 2, \ldots)$ and at failure. The expected cost $B(T)$ per unit of time following a block replacement policy at interval T over an infinite time span is given by

$$B(T) = \frac{C_1 M(T) + C_2}{T} \tag{1}$$

where $M(T)$ is the renewal function (expected number of failures in $[0, T]$) corresponding to the underlying failure distribution $F(T)$. We know from reference [1] that a necessary condition that a finite value of T_0 minimizes $B(T)$ is that it satisfies equation (2) obtained by differenciating equation (1) and setting the derivative equal to zero:

$$\left.\frac{\mathrm{d}B}{\mathrm{d}T}\right|_{T=T_0} = 0 \tag{2}$$

$$\Rightarrow T_0 m(T_0) - M(T_0) = \frac{C_2}{C_1}$$

where $m(T_0)$ is the derivative of $M(T_0)$ and is known as renewal density.

$M(T)$ can be written as follows:

$$M(T) = \int_0^T [1 + M(T-x)] f(x) \mathrm{d}x \tag{3}$$

Now let

$$[1 + M(T-x)] f(x) = g(T, x)$$

We know that,

$$\left[\int_0^T g(T, x) \mathrm{d}x\right](t) = \left[\int_0^T \partial_1 g(T, x) \mathrm{d}x\right] + g(T, T_0)$$

Now

$$g(T, x) = f(x) + M(T-x) f(x)$$

and

$$\partial_1 g(T, x) = f(x) m(T-x)$$

Thus for $m(T)$ we have,

$$m(T) = \left[\int_0^T f(x) m(T-x) \mathrm{d}x\right] + [f(T) + M(T-T_0) f(T)] =$$

$$= \int_0^T f(x) m(T-x) \mathrm{d}x + f(T) \tag{4}$$

It is to be noted that reference [1, page 50] carries an error probably a printing error where $m(T)$ is calculated as

$$m(T)=\int_0^T [1+m(T-x)]f(x)\mathrm{d}x,$$

which is quite wrong.

In section 3 optimum values of block time for different values of $\frac{C_2}{C_1}$ are calculated. Fig. 1 shows that not all the values of T_0 derived with the help of the equation 2 are optimum.



Fig. 1. Block replacement policy. (T_b = block replacement interval, PM = preventive maintenance, RF = restoration at failure)

For the derivative of $\frac{\mathrm{d}B}{\mathrm{d}T}$, we get the following equation:

$$\frac{\mathrm{d}^2B}{\mathrm{d}T^2}=\frac{C_1[m(T)+Tm'(T)-m(T)]T^2-2T\{C_1[Tm(T)-M(T)]-C_2\}}{T^4} \tag{5}$$

Now if we want to determine the optimum block time T we get from equation (5) that the following condition should also be satisfied:

$$\frac{C_1m'(T_0)}{T_0}>0 \tag{6}$$

that is $\operatorname{sgn} m'(T_0)=1$.

We know from trapezoidal formula that:

$$\int_0^T f(x)\mathrm{d}x\approx\frac{h}{2}\sum_{i=0}^{N}\alpha_i^N f(a+ih)$$

where,

$$h=\frac{b-a}{N} \quad \text{and} \quad \alpha_i^N=\begin{cases}1 & i=N \quad \text{or} \quad i=0\\ 2 & \text{otherwise}\end{cases}$$

Using the above formula and after mathematical simplification what we get from $M(T)$ is:

$$\text{For} \quad i=1 \qquad M(1h)=\frac{F(1h)}{1-\frac{h}{2}f(0)} \tag{7.a}$$

and $i=2,\ldots,N$

$$M(ih)=\frac{F(ih)+h\sum_{j=1}^{i-1} f(h(i-j))M(jh)}{1-\frac{h}{2}f(0)} \tag{7.b}$$

And for $m(T)$ we get the following expression:

$$\text{For} \quad i=1 \qquad m(1h)=\frac{f(1h)}{1-\frac{h}{2}f(0)} \tag{8.a}$$

and for $i=2,3,\ldots,N$,

$$m(ih)=\frac{f(ih)+h\sum_{j=1}^{i-1} f(h(i-j))m(jh)}{1-\frac{h}{2}f(0)} \tag{8.b}$$

For the purpose of iteration we express equation (2) as follows:

$$t_{n+1}=t_n-\frac{(t_n m(t_n)-M(t_n)-C_{21})(t_n-t_{n-1})}{t_n(m(t_n)-m(t_{n-1}))}$$

where $C_{21}=\frac{C_2}{C_1}$

If $\xi(T)=Tm(T)-M(T)-C_{21}=0$

Then
$$t_{n+1}=t_n-\frac{\xi(t_n)(t_n-t_{n-1})}{[m(t_n)-m(t_{n-1})]t_n}. \tag{9}$$

By using a computer programme written in BASIC language in COMMODORE 64 the optimum value of T_0 is determined. Only those values of T_0 are accepted which satisfies the necessary condition (6) that is $\frac{C_1 \cdot m'(T_0)}{T_0} > 0$.

3. Simplification of the determination of preventive maintenance interval for different values of $\frac{C_2}{C_1}$

$$\text{let} \quad \frac{C_2}{C_1} = C(T_0)$$

Then equation (2) can be written as follows:

$$T_0 \cdot m(T_0) - M(T_0) = C(T_0)$$

We use the same computer programme (list of which is given in the appendix) mentioned in section 2, for different values of T_0, the values of $M(T_0)$ and $m(T_0)$ are calculated. By using these values f T_0, $M(T_0)$ and $m(T_0)$, the values of $C(T_0)$ is determined for given T_0. Thus we get Fig. 2 that enables us to read optimum values of T_0 for different values of $\frac{C_2}{C_1}$.

Example

Suppose that $F(t) = 1 - e^{-at^b}$ where $t > 0$ and $a = 1.5$ (scale parameter of a Weibull distribution) $b = 2$ (shape parameter of a Weibull distribution).

We are interested in $C(T_0)$. Now by using the computer programme given in the appendix we get Fig. 2, from which for different values of $\frac{C_2}{C_1}$ the optimum block time T_0 can be easily read, for given a and b.

4. Conclusion

An error committed in [1] for calculating the renewal density function is corrected. A necessary condition for calculating the optimum block interval is derived. In real life the diagram shown in Fig. 2 may be used to determine the modified value of T_0 when $\frac{C_2}{C_1}$ is changed. With the help of the procedure discussed in section 2 the block time can be easily ascertained.



Fig. 2. Block diagram of the computer program used for the generation of failure distribution under preventive maintenance. In the figure the following symbols are used of which the respective connotations are given here:

F: Name of the file where the generated failure data are to be stored
NM: Number of the failures which could not be prevented
TN-D: Time between two consecutiv failures that could not be prevented
TB: Optimum block interval
D, TP: Program variables
INT(TN/TB): Number of cycle in which the failure has occurred

Reference

1. Barlow, R. E.—Proschan, F.: Mathematical Theory of Reliability, Wiley, New York, p. 256 (1965).

Chandan Chowdhury P. O. Tinsukia Tambulbari
Assam PIN-786125
India

Appendix

```
10000 REM **** FEEDING OF DATA ****
10010 OPEN1,4
10020 PRINT
10030 PRINT "      ,------------------------------,"
10040 PRINT "      ! CHOOSE DENSITY FUNCTION !   !"
10050 PRINT "      '------------------------------'"
10060 PRINT "                         WEIBULL"
10070 PRINT "                         LOGNORMAL"
10080 PRINT "                         GAMMA "
10090 GET E$:IF E$<>"W" AND E$<>"L" AND E$<>"G" THEN 10090
10100 PRINT
10110 PRINT "         ,----------------------,"
10120 PRINT "         !  GIVE PARAMETERS !  !"
10130 PRINT "         '----------------------'"
10140 IF E$="W" THEN 10170
10150 IF E$="L" THEN 10230
10160 IF E$="G" THEN 10290
10170 PRINT
10180 INPUT "     SCALE PARAMETER (A)=  ";A :PRINT#1,"  A      =";A
10190 INPUT "     SHAPE PARAMETER (B)=  ";B :PRINT#1,"  B      =";B
10195 REM LF:DENSITY FUNCTION
10200 DEF FNLF(X)=A*B*X↑(B-1)*EXP(-A*X↑B)
10205 REM GF:FAILURE DISTRIBUTION
10210 DEF FNGF(X)=1-EXP(-A*X↑B)
10220 GOTO 10380
10230 PRINT
10235 REM S,M ARE THE PARAMETERS OF A LOGNORMAL DISTRIBUTION
10240 INPUT "     SIGMA =   ";S :PRINT#1,"  SIGMA =";S
10250 INPUT "     MU    =   ";M :PRINT#1,"  MU    =";M
10260 E=SQR(2*π)
10270 DEF FNLF(X)=EXP(-(LOG(X)-M)↑2/(2*S↑2))/(X*S*E)
10280 GOTO 10380
10290 PRINT
10295 REM AL,LA ARE THE PARAMETERS OF A GAMMA DISTRIBUTION
10300 INPUT "     ALPHA =   ";AL:PRINT#1,"  ALPHA =";AL
10310 INPUT "     LAMBDA=   ";LA:PRINT#1,"  LAMBDA=";LA
10320 A=1/12:B=1/288:C=139/51840:D=571/2488320:E=SQR(2*π)
10330 X=AL:Z=X+5
10340 Y=E*Z↑(Z-1/2)/EXP(Z)*(1+A/Z+B/Z↑2-C/Z↑3-D/Z↑4)
10350 Y=Y/(X*(X+1)*(X+2)*(X+3)*(X+4))
10360 CC=LA↑AL/Y
10370 DEF FNLF(X)=CC*X↑(AL-1)*EXP(-LA*X)
10375 REM EP:ACCURACY OF THE OPTIMUM BLOCK TIME
10380 INPUT "     EPS   =   ";EP:PRINT#1,"  EPS   =";EP
10385 REM TM:UPPER LIMIT OF TIME TO BE CONSIDERED
10390 INPUT "     TMAX  =   ";TM:PRINT#1,"  TMAX  =";TM
10395 REM TM:NUMBER OF DISCRETE TIME INTERVALS FOR NUMERICAL TREATMENT
10400 INPUT "     N     =   ";N :PRINT#1,"  N     =";N
10410 PRINT#1
10420 H=TM/N
10430 DIM LF(N),GF(N),GM(N),LM(N),U(N)
10440 IF E$<>"L" THEN LF(0)=FNLF(0)
10450 FOR I=1 TO N
10460 LF(I)=FNLF(I*H)
10470 NEXT I
10480 IF E$="W" THEN 10580
10485 REM **** COMPUTATION OF 'GF' IN CASE OF LOGNORMAL AND GAMMA DIST. ****
10490 GF(1)=(LF(0)+LF(1))*H/2
10500 FOR I=2 TO N
10510 W=(LF(0)+LF(I))*H/2
10520 FOR J=1 TO I-1
10530 W=W+LF(J)*H
10540 NEXT J
```

```
10550 GF(I)=W
10560 NEXT I
10570 GOTO 10610
10575 REM **** COMPUTATION OF 'GF' IN CASE OF WEIBULL DIST. ****
10580 FOR I=0 TO N
10590 GF(I)=FNGF(I*H)
10600 NEXT I
10605 REM **** COMPUTATION OF 'LM' AND 'GM' ****
10610 REM GM:RENEWAL FUNCTION
10612 REM LM:RENEWAL DENSITY FUNCTION
10615 D=1-H/2*LF(0)
10620 GM(1)=GF(1)/D
10630 FOR I=2 TO N
10640 W=0
10650 FOR J=1 TO I-1
10660 W=W+LF(I-J)*GM(J)
10670 NEXT J
10680 GM(I)=(GF(I)+H*W)/D
10690 NEXT I
10700 LM(1)=LF(1)/D
10710 FOR I=2 TO N
10720 W=0
10730 FOR J=1 TO I-1
10740 W=W+LF(I-J)*LM(J)
10750 NEXT J
10760 LM(I)=(LF(I)+H*W)/D
10770 NEXT I
10780 FOR I=0 TO N
10790 U(I)=H*I*LM(I)-GM(I)
10800 NEXT I
10810 FOR I=0 TO N
10815 REM **** PRINTING ****
10820 PRINT#1,I;"       T=",I*H;"       C2/C1=";U(I)
10830 NEXT I
10840 PRINT#1
10843 REM **** INPUT COSTS ****
10845 REM C1 COST DUE TO FAILURE
10846 REM C2 COST OF PREVENTIVE MAINTENANCE
10850 INPUT "    C1     =    ";C1:PRINT#1,"  C1     =";C1
10860 INPUT "    C2     =    ";C2:PRINT#1,"  C2     =";C2
10870 PRINT#1
10880 CLOSE1
10885 Q=C2/C1:IF Q<0 THEN 11210
10890 REM **** GRAPHICAL REPRESENTATION ****
10900 MX=0:DX=0
10910 FOR I=0 TO N
10920 IF ABS(U(I))>MX THEN MX=ABS(U(I)):DX=I
10930 NEXT I
10940 HIRES 6,3
10950 FOR I=0 TO DX-1
10960 Y0=199-(U(I))/MX*195
10970 Y1=199-(U(I+1))/MX*195
10980 LINE 319/N*I,Y0,319/N*(I+1),Y1,1
10990 NEXT I
11000 Z=199-Q/MX*195
11010 LINE 0,Z,319,Z,1
11015 LINE 0,4,0,199,1
11020 GET E$:IF E$="" THEN 11020
11030 IF Q=0 THEN COPY
11040 OPEN1,4
11050 NRM
11054 REM **** ITERATION FOR ASCERTAINING OPTIMUM BLOCK TIME ****
11055 REM T0:INITIAL VALUE FOR ITERATION
11060 INPUT "    T0     =    ";T0:PRINT#1,"  T0     =";T0
11070 TP=T0
11080 T =T0-EP
11090 X=T0:GOSUB 11220:LP=LM
```

```
11100 X=T :GOSUB 11220
11110 R=(T-TP)/T*(T*LM-GM-Q)/(LM-LP)
11120 R=T-R
11130 TP=T:T=R:LP=LM
11140 PRINT "    T=";T
11150 IF ABS(T-TP)<TP*1E-7 THEN 11170
11160 GOTO 11100
11170 X=T:GOSUB 11220
11180 PRINT#1,"OPTIMUM BLOCK TIME :   T=",T
11190 PRINT#1,"EXPECTED COST PER UNIT TIME :   B(T)=";(C1*GM+C2)/T
11200 GOTO 10840
11210 END
11215 REM **** LINEAR INTERPOLATION OF 'LM' AND 'GM' ****
11220 XH=X/H:I=INT(XH)
11230 GM=(I+1-XH)*GM(I)+(XH-I)*GM(I+1)
11240 LM=(I+1-XH)*LM(I)+(XH-I)*LM(I+1)
11250 RETURN

READY.
```